Trigonometrikus függvények integrálása

Természetes kitevőjű hatványok

1) $\int \sin(x)^n \, dx$, $\int \cos(x)^n \, dx$, $\int \sin(x)^n \cos(x)^m \, dx$

integrálása. (n,m természetes szám)

A sin(x) és a cos(x) integrálja alapintegrál. Legyen először **n = 2.** Ekkor az un. linearizáló formulát alkalmazzuk. Amint ezeket már levezettük:

$$\cos(x)^2 = \frac{1}{2} + \frac{\cos(2x)}{2}$$

$$\sin(x)^2 = \frac{1}{2} - \frac{\cos(2x)}{2}$$

```
> Int(cos(x)^2,x)=Int(1/2+1/2*cos(2*x),x);
```

$$\int \cos(x)^2 \, dx = \int \frac{1}{2} + \frac{1}{2}\cos(2x) \, dx \qquad (1.1)$$

```
> Int(cos(x)^2,x)=int(1/2+1/2*cos(2*x),x);
```

$$\int \cos(x)^2 \, dx = \frac{1}{2}x + \frac{1}{4}\sin(2x) \qquad (1.2)$$

```
> Int(sin(x)^2,x)=Int(1/2-1/2*cos(2*x),x);
```

$$\int \sin(x)^2 \, dx = \int \frac{1}{2} - \frac{1}{2}\cos(2x) \, dx \qquad (1.3)$$

```
> Int(sin(x)^2,x)=int(1/2-1/2*cos(2*x),x);
```

$$\int \sin(x)^2 \, dx = \frac{1}{2}x - \frac{1}{4}\sin(2x) \qquad (1.4)$$

α) **Ha n>2 és páros**, akkor az elsõ két esetben vagy a linearizáló formula ismételt alkalmazásával, vagy parciális integrálás segítségével boldogulhatunk. A harmadik integrál a $\sin(x)^2 + \cos(x)^2 = 1$ összefüggés segítségével az elõzõk valamelyikére vezethetõ vissza. **Példa:**

```
> sin(x)^4=(combine(sin(x)^2))^2;
```

$$\sin(x)^4 = \left(\frac{1}{2} - \frac{1}{2}\cos(2x)\right)^2 \qquad (1.5)$$

```
> sin(x)^4=combine(rhs(%),trig);
```

$$\sin(x)^4 = \frac{3}{8} - \frac{1}{2}\cos(2x) + \frac{1}{8}\cos(4x) \qquad (1.6)$$

```
> Int(sin(x)^4,x)=Int(3/8+1/8*cos(4*x)-1/2*cos(2*x),x);
```

$$\int \sin(x)^4\,dx = \int \frac{3}{8} - \frac{1}{2}\cos(2x) + \frac{1}{8}\cos(4x)\,dx \tag{1.7}$$

```
> Int(sin(x)^4,x)=int(3/8+1/8*cos(4*x)-1/2*cos(2*x),x);
```

$$\int \sin(x)^4\,dx = \frac{3}{8}x - \frac{1}{4}\sin(2x) + \frac{1}{32}\sin(4x) \tag{1.8}$$

A parciális integrálás segítségével *rekurzív formulát* nyerhetünk a $\sin(x)^n$,$\cos(x)^n$ ill. $\tan(x)^n$ integálására.
Lássuk példaként a $\sin(x)^n$ integrálását!

```
> with(student):# A student csomag tartalmazza a szükséges
  eljárásokat
> Int(sin(x)^n,x)=simplify(intparts(Int(sin(x)^n,x),sin(x)^(n-1)))
  ;
```

$$\int \sin(x)^n\,dx = -\sin(x)^{(n-1)}\cos(x) + \int \sin(x)^{(n-2)}\cos(x)^2\,dx\,n - \int \sin(x)^{(n-2)}\cos(x)^2\,dx \tag{1.9}$$

```
> lhs(%)=op(1,rhs(%))+simplify(op(2,rhs(%))+op(3,rhs(%)));
```

$$\int \sin(x)^n\,dx = -\sin(x)^{(n-1)}\cos(x) + \int \sin(x)^{(n-2)}\cos(x)^2\,dx\,(n-1) \tag{1.10}$$

```
> lhs(%)=subs(cos(x)^2=1-sin(x)^2,rhs(%));
```

$$\int \sin(x)^n\,dx = -\sin(x)^{(n-1)}\cos(x) + \int \sin(x)^{(n-2)}\left(1-\sin(x)^2\right)dx\,(n-1) \tag{1.11}$$

```
> lhs(%)=op(1,rhs(%))+expand(op(2,rhs(%)),trig);
```

$$\int \sin(x)^n\,dx = -\sin(x)^{(n-1)}\cos(x) + \int \frac{\sin(x)^n}{\sin(x)^2}\,dx\,n - \int \frac{\sin(x)^n}{\sin(x)^2}\,dx - \int \sin(x)^n\,dx\,n + \int \sin(x)^n\,dx \tag{1.12}$$

```
> lhs(%)=simplify(solve(%,Int(sin(x)^n,x)));
```

$$\int \sin(x)^n\,dx = \frac{-\sin(x)^{(n-1)}\cos(x) + \int \sin(x)^{(n-2)}\,dx\,n - \int \sin(x)^{(n-2)}\,dx}{n} \tag{1.13}$$

Innen látható, hogy $I_n = \int \sin(x)^n\,dx$ **jelöléssel:**

$$I_n = -\frac{\sin(x)^{(n-1)}\cos(x)}{n} + \frac{(n-1)\,I_{n-2}}{n}$$

```
> intsin:=proc(n::posint)
  option remember:
  if n=1 then RETURN(-cos(x)) fi:
  if n=2 then RETURN( x/2-sin(2*x)/4) fi:
  -sin(x)^(n-1)*cos(x)/n+(n-1)*intsin(n-2)/n:
  end:
```

Számítsuk ki az eljárással a $\int sin(x)^3\, dx$ értékét

```
> intsin(3);
```

$$-\frac{1}{3}\sin(x)^2\cos(x) - \frac{2}{3}\cos(x) \tag{1.14}$$

Ellenőrizzük az eredményt driválással:

```
> diff(%,x);
```

$$-\frac{2}{3}\sin(x)\cos(x)^2 + \frac{1}{3}\sin(x)^3 + \frac{2}{3}\sin(x) \tag{1.15}$$

Hozzuk egyszerűbb alakra:

```
> simplify(%);
```

$$\sin(x)^3 \tag{1.16}$$

β) Ha a kitevõ vagy a kitevõk valamelyike **páratlan**, akkor $f^{\alpha}\,\mathbf{f}'$ tudunk elõállítani. Például:

```
> Int(sin(x)^3*cos(x)^2,x)=Int(sin(x)*(1-cos(x)^2)*cos(x)^2,x);
```

$$\int \sin(x)^3\cos(x)^2\,dx = \int \sin(x)\left(1-\cos(x)^2\right)\cos(x)^2\,dx \tag{1.17}$$

```
> Int(sin(x)*(cos(x)^2-cos(x)^4),x)=int(sin(x)*(1-cos(x)^2)*cos(x)
  ^2,x);
```

$$\int \sin(x)\left(\cos(x)^2-\cos(x)^4\right)dx = \frac{1}{5}\cos(x)^5 - \frac{1}{3}\cos(x)^3 \tag{1.18}$$

Trigonometrikus szorzatok összeggé alakítása

2. A $\sin(n\,x)\sin(m\,x)$, $\sin(m\,x)\cos(n\,x)$ és $\cos(m\,x)\cos(n\,x)$ alakú kifejezések integrálása.
Ezeket a kifejezéseket összeggé alakíthatjuk:(Lásd: a Függvénytáblázatban)

```
> sin(n*x)*sin(m*x) = combine(sin(n*x)*sin(m*x),trig);
```

$$\sin(n\,x)\sin(m\,x) = \frac{1}{2}\cos(n\,x-m\,x) - \frac{1}{2}\cos(n\,x+m\,x) \tag{2.1}$$

```
> sin(n*x)*cos(m*x) = combine(sin(n*x)*cos(m*x),trig);
```

$$\sin(n\,x)\cos(m\,x) = \frac{1}{2}\sin(n\,x+m\,x) + \frac{1}{2}\sin(n\,x-m\,x) \quad (2.2)$$

```
> cos(n*x)*cos(m*x) = combine(cos(n*x)*cos(m*x),trig);
```

$$\cos(n\,x)\cos(m\,x) = \frac{1}{2}\cos(n\,x-m\,x) + \frac{1}{2}\cos(n\,x+m\,x) \quad (2.3)$$

```
> Int(sin(n*x)*sin(m*x),x)=Int(1/2*cos(n*x-m*x)-1/2*cos(n*x+m*x),
  x);
```

$$\int \sin(n\,x)\sin(m\,x)\,\mathrm{d}x = \int \frac{1}{2}\cos(n\,x-m\,x) - \frac{1}{2}\cos(n\,x+m\,x)\,\mathrm{d}x \quad (2.4)$$

```
> Int(sin(n*x)*sin(m*x),x)=int(1/2*cos(n*x-m*x)-1/2*cos(n*x+m*x),
  x);
```

$$\int \sin(n\,x)\sin(m\,x)\,\mathrm{d}x = \frac{1}{2}\,\frac{\sin((n-m)\,x)}{n-m} - \frac{1}{2}\,\frac{\sin((n+m)\,x)}{n+m} \quad (2.5)$$

Szögfüggvények racionális függvényei

Trigonometrikus függvények racionális függvényeinek integrálására minden esetben alkalmazható a

$$\boldsymbol{t = \tan\left(\frac{x}{2}\right)}$$

helyettesítés

Valamennyi szögfüggvény kifejezhetõ a $\tan\left(\frac{x}{2}\right)$ racionális függvényeként:

$$\sin(x) = \frac{2\sin\left(\frac{x}{2}\right)\cos\left(\frac{x}{2}\right)}{\sin\left(\frac{x}{2}\right)^2+\cos\left(\frac{x}{2}\right)^2} = \frac{2\tan\left(\frac{x}{2}\right)}{\tan\left(\frac{x}{2}\right)^2+1} = \frac{2\,t}{t^2+1}.$$

Tehát: $\boldsymbol{\sin(x) = \frac{2\,t}{t^2+1}}$

Hasonlóan nyerhetõ, hogy:

$$\cos(x) = \frac{1-t^2}{1+t^2}$$

A dt differenciálra pedig:

```
> t=tan(x/2);
```

$$t = \tan\left(\frac{1}{2}\,x\right) \quad (3.1)$$

```
> isolate(%,x);
```

$$x = 2\arctan(t) \quad (3.2)$$

```
> dx=diff(rhs(%),t)*dt;
```

$$dx = \frac{2\,dt}{t^2+1} \tag{3.3}$$

Példa:

$$\int \frac{1}{\sin(x)}\,\mathrm{d}x = \int \frac{2}{\frac{(1+t^2)\,2\,t}{1+t^2}}\,\mathrm{d}t = \int \frac{1}{t}\,\mathrm{d}t = \ln(|t|) = \ln\left(\left|\tan\left(\frac{x}{2}\right)\right|\right)$$

```
> x:='x':
> Int(1/sin(x),x)=changevar(tan(x/2)=t,Int(1/sin(x),x),t);
```

$$\int \frac{1}{\sin(x)}\,\mathrm{d}x = \int \frac{2}{\sin(2\arctan(t))\,(t^2+1)}\,\mathrm{d}t \tag{3.4}$$

```
> lhs(%)=Int(map(z->expand(z),op(rhs(%))),t);
```

$$\int \frac{1}{\sin(x)}\,\mathrm{d}x = \int \frac{1}{t}\,\mathrm{d}t \tag{3.5}$$

```
> lhs(%)=value(rhs(%));
```

$$\int \frac{1}{\sin(x)}\,\mathrm{d}x = \ln(t) \tag{3.6}$$

```
> lhs(%)=subs(t=tan(x/2),rhs(%));
```

$$\int \frac{1}{\sin(x)}\,\mathrm{d}x = \ln\left(\tan\left(\frac{1}{2}\,x\right)\right) \tag{3.7}$$

```
>
```